Solutions to Exercise: Shaper with Unequal Corner Frequencies

PF@SuS@UniHD 2016+

We have $r := \frac{\omega_l}{\omega_h} = \frac{\tau_h}{\tau_l}$

Some Defaults

In[122]:=
```
Clear[k, T, R, C];
```
Clear: Symbol C is Protected.

In[123]:=
```
$Assumptions = r > 0 && τ > 0 && τh > 0 && τl > 0 &&
   ω > 0 && ωh > 0 && ωl > 0 && B > 0 && s > 0 && -3 < Re[k] < 1 && ωCRRC > 0;
```

In[124]:=
```
SetOptions[{LogLogPlot, LogLinearPlot, Plot},
  {Frame → True, Filling → Axis, ImageSize → 400, PlotLegends → "Expressions"}];
```

1. Transfer Function

In[125]:= $H[s\_] = \frac{1}{1 + s\,\tau_l}\,\frac{s\,\tau_h}{1 + s\,\tau_h}$ /. $\tau_l \to \frac{\tau_h}{r}$ /. $\tau_h \to \tau$

Out[125]= $\frac{s\,\tau}{(1 + s\,\tau)\left(1 + \frac{s\,\tau}{r}\right)}$

```
Gain[s_, τ_, r_] = Abs[H[s]] // Simplify (* τ is corner frequency of high pass,
  r is ratio between high and low pass *)
```

Out[126]= $\frac{r\,s\,\tau}{(1 + s\,\tau)\,(r + s\,\tau)}$

For r = 1, we have the normal CR-RC filter with equal corners.
For r < 0, the $\tau_l > \tau_h$, or $\omega_l < \omega_h$which is stupid: the low pass starts to drop the signal before the high pass
r>1 is the intersting case where we have a ‘trapezoidal’ filter shape with a flat plateau.

In[130]:=
```
LogLogPlot[{Gain[s, 1, 100000], Gain[s, 1, 1000], Gain[s, 1, 1], Gain[s, 1, 0.01]},
 {s, 0.001, 1000000}
 , PlotRange → {{0.001, 1000000}, {0.001, 2}}, GridLines → {{1}, {1, 1/4}}]
```

Out[130]=



2. Step Reponse

In[131]:=
```
Clear[v];
v[t_, τ_, r_] = InverseLaplaceTransform[1/s H[s], s, t] // Simplify
```

Out[132]=
$$-\frac{\left(e^{-\frac{t}{\tau}} - e^{-\frac{r\,t}{\tau}}\right) r}{1 - r}$$

In[133]:=
```
Plot[{v[t, 1, 5], v[t, 1, 0.99], v[t, 1, 0.2]}, {t, 0, 5}]
```

Out[133]=



3. Peaking Time and Amplitude

In[134]:=
```
D[v[t, τ, r], t] // Simplify
```

Out[134]=
$$\frac{r\left(e^{-\frac{t}{\tau}} - e^{-\frac{r\,t}{\tau}}\, r\right)}{\tau - r\,\tau}$$

In[135]:=
```
Clear[tpeak];
tpeak[τ_, r_] = t /. Solve[D[v[t, τ, r], t] == 0, t] // First // Simplify
```

Solve: Inverse functions are being used by Solve, so some solutions may not be found; use Reduce for complete solution information.

Out[135]= $\frac{\tau \, \text{Log}[r]}{-1 + r}$

In[137]:=
```
vpeak[r_] = v[tpeak[τ, r], τ, r] // Simplify
```

Out[137]= $r^{\frac{1}{1-r}}$

This vpeak is a funy formula. What is the limit for this for r->1 ??? It shouls be our well known 1/e for the CR-RC shaper...

In[142]:= `Limit[vpeak[r], r → 1] ==` $\frac{1}{e}$

Out[142]= True

In[143]:=
```
LogLinearPlot[tpeak[1, r], {r, 0.1, 10}]
```

Out[143]=


```
Plot[{v[t, 1, 5], v[t, 1, 0.99], v[t, 1, 0.2]}, {t, 0, 6}, PlotRange → {0, 0.8},
 Epilog → {PointSize[Large], Brown
    , Point[{tpeak[1, 5], vpeak[1, 5]}]
    , Point[{tpeak[1, 0.99], vpeak[1, 0.99]}]
    , Point[{tpeak[1, 0.2], vpeak[1, 0.2]}]}
]
```

Out[145]=



4. Simple Shaper Limit (r->1)

In[146]:= `Limit[{v[t, τ, r], tpeak[τ, r], vpeak[τ, r]}, r → 1]`

Out[146]= $\left\{\frac{e^{-\frac{t}{\tau}}\, t}{\tau},\ \tau,\ \frac{1}{e}\right\}$

5. Noise Integral

In[147]:= `H2[ω_] = H[i ω] Conjugate[H[i ω]] // Simplify`

Out[147]= $\frac{r^2\,\tau^2\,\omega^2}{\left(1+\tau^2\,\omega^2\right)\,\left(r^2+\tau^2\,\omega^2\right)}$

In[149]:=

```
INT[k_] = Integrate[ω^k H2[ω], {ω, 0, ∞}] // FullSimplify
(* This is the hard integral *)
(* Note that we have restricted k in the $Assumptions to -3<Re[k]<1,
so that the expressions get simpler. If we do not do that,
we get conditional expressions *)
```

Out[149]= $\frac{\pi\, r^2\,\left(-1+r^{1+k}\right)\,\tau^{-1-k}\,\mathrm{Sec}\left[\frac{k\,\pi}{2}\right]}{2\,\left(-1+r^2\right)}$

In[150]:= `Coeff = Table[Limit[INT[k], k → K], {K, -2, 0}] // Simplify`

Out[150]= $\left\{\frac{\pi\, r\,\tau}{2+2\,r},\ \frac{r^2\,\mathrm{Log}[r]}{-1+r^2},\ \frac{\pi\, r^2}{2\,\tau+2\,r\,\tau}\right\}$

```
CoefLatex =
  Table[Limit[1/2 ω^(k+1) (r^(k+1) - 1)/(r^2 - 1) Gamma[(1 + k)/2] Gamma[(1 - k)/2] /. ω → 1/(r τ), k → K], {K, -2,
     0}] /. r → 1 / r // Simplify(* For PF: Compare to old LaTeX solution *)
```

Out[151]= $\left\{\frac{\pi\, r\, \tau}{2+2\, r},\ \frac{r^2\, \text{Log}[r]}{-1+r^2},\ \frac{\pi\, r^2}{2\, \tau+2\, r\, \tau}\right\}$

In[152]:= `Coeff == CoefLatex (* Check that both are the same *)`

Out[152]= True

6. SNR for same peaking time

In[153]:=
```
ttrans = Solve[tpeak[τ, r] == tpref, τ] // First
  (* Find the τ we need in the general case to keep peaking time at tpref *)
```

Out[153]= $\left\{\tau \to \frac{(-1+r)\ \text{tpref}}{\text{Log}[r]}\right\}$

In[154]:=
```
XX = INT[k] / vpeak[τ, r]^2 /. ttrans /. tpref → 1/ω_CRRC // FullSimplify
  (* XX is INT[] scaled for same amplitude and peaking *)
```

Out[154]= $\frac{\pi\, r^{\frac{2 r}{-1+r}}\ \left(-1+r^{1+k}\right)\ \text{Sec}\left[\frac{k\,\pi}{2}\right]\ \left(\frac{-1+r}{\text{Log}[r]\ \omega_{\text{CRRC}}}\right)^{-1-k}}{2\ \left(-1+r^2\right)}$

In[155]:= `Limit[XX, r → 1] (* Lecture case for normal CR-RC-shaper *)`

Out[155]= $\frac{1}{4}\ e^2\ (1+k)\ \pi\ \text{Sec}\left[\frac{k\,\pi}{2}\right]\ \omega_{\text{CRRC}}^{1+k}$

In[157]:= `Table[Limit[XX, k → K], {K, -2, 0}] // FullSimplify`

Out[157]= $\left\{\text{ConditionalExpression}\left[\frac{\pi\ (-1+r)\ r^{\frac{1+r}{-1+r}}}{2\ (1+r)\ \text{Log}[r]\ \omega_{\text{CRRC}}},\ (-1+r)\ \text{Re}\left[\frac{1}{\text{Log}[r]}\right]>0\right],\right.$
$\text{ConditionalExpression}\left[\frac{r^{\frac{2 r}{-1+r}}\ \text{Log}[r]}{-1+r^2},\ (-1+r)\ \text{Re}\left[\frac{1}{\text{Log}[r]}\right]>0\right],$
$\left.\text{ConditionalExpression}\left[\frac{\pi\, r^{\frac{2 r}{-1+r}}\ \text{Log}[r]\ \omega_{\text{CRRC}}}{2\ \left(-1+r^2\right)},\ (-1+r)\ \text{Re}\left[\frac{1}{\text{Log}[r]}\right]>0\right]\right\}$

For the moment, I ignore the conditional expression...

In[158]:= `Limit[%, r → 1]`

Out[158]= $\left\{\frac{e^2\,\pi}{4\ \omega_{\text{CRRC}}},\ \frac{e^2}{2},\ \frac{1}{4}\ e^2\ \pi\ \omega_{\text{CRRC}}\right\}$

In[159]:= $\% == \dfrac{A^2\ \frac{\pi}{4} \left\{\frac{1}{\omega_{CRRC}},\ \frac{2}{\pi},\ \omega_{CRRC}\right\}}{(A / e)^2}$ `(* result from lecture slides p.27 *)`

Out[159]= True

7. Find Minimum

In[160]:=
```
DXX = D[XX, r] // FullSimplify
```

Out[160]= $\dfrac{1}{2\ (-1+r)^4\ (1+r)^2} \pi\ r^{\frac{1+r}{-1+r}} \left(\dfrac{-1+r}{\mathrm{Log}[r]}\right)^{-k} \left((1+k)\ (-1+r)^2\ (1+r)\ \left(-1+r^{1+k}\right) + r\ \mathrm{Log}[r]\ \left((-1+r)\ \left(-1+k+r+k\,r-r^k\ (1+k+(-1+k)\ r)\right) - 2\ (1+r)\ \left(-1+r^{1+k}\right)\ \mathrm{Log}[r]\right)\right)\ \mathrm{Sec}\left[\dfrac{k\,\pi}{2}\right]\ \omega_{CRRC}^{1+k}$

In[161]:=
```
Limit[DXX, r → 1] // FullSimplify
 (* The derivative of XX is zero at r=1 → CR-RC has minimal noise for ALL k *)
```

Out[161]= 0

In[162]:=
```
PP = Table[Limit[XX /. ωCRRC → 1, k → K], {K, -2, 0}] // Simplify
  (* Make an expression with numerical values for plotting *)
```

Out[162]= $\left\{\mathrm{ConditionalExpression}\left[\dfrac{\pi\ (-1+r)\ r^{\frac{1+r}{-1+r}}}{2\ (1+r)\ \mathrm{Log}[r]},\ (-1+r)\ \mathrm{Re}\left[\dfrac{1}{\mathrm{Log}[r]}\right] > 0\right],\right.$
$\mathrm{ConditionalExpression}\left[\dfrac{r^{\frac{2r}{-1+r}}\ \mathrm{Log}[r]}{-1+r^2},\ (-1+r)\ \mathrm{Re}\left[\dfrac{1}{\mathrm{Log}[r]}\right] > 0\right],$
$\left.\mathrm{ConditionalExpression}\left[\dfrac{\pi\ r^{\frac{2r}{-1+r}}\ \mathrm{Log}[r]}{2\ \left(-1+r^2\right)},\ (-1+r)\ \mathrm{Re}\left[\dfrac{1}{\mathrm{Log}[r]}\right] > 0\right]\right\}$

In[163]:=
```
LogLogPlot[Evaluate[PP], {r, 0.01, 100}, Filling → Axis, PlotRange → {3, 7}]
```

Out[163]=



In[164]:=
```
Limit[PP, r → 1]
```

Out[164]= $\left\{\dfrac{e^2\,\pi}{4},\ \dfrac{e^2}{2},\ \dfrac{e^2\,\pi}{4}\right\}$